Reading STEREO ephemerides as SPICE kernels

William Thompson March 9, 2005

1 Introduction

The STEREO orbit and attitude data will be distributed as SPICE kernel files. Many members of the STEREO team may not be familiar with the SPICE library. Therefore, this document is intended to serve as an introduction to the SPICE system, and how it applies to STEREO.

As of this initial writing, only a sample of the STEREO ephemeris is available. Therefore, for the moment, we will only address the question of extracting the positions of the spacecraft. We will extend this introduction to include orientation when sample attitude files are also available.

SPICE is available as both FORTRAN and C libraries. The CSPICE library can also be used from IDL as a Dynamically Loadable Module (DLM). We will concentrate on the IDL interface, called Icy, but the discussion is also applicable to the FORTRAN and C libraries with little modification.

2 Installing SPICE

The SPICE software and associated data files are available via the web from

```
http://naif.jpl.nasa.gov/naif/
```

Follow the "Toolkit" link to download the software based on the software environment (e.g. IDL) and the operating system (e.g. Linux). The files can be unpacked into a suitable location, for example /usr/local/icy. No compilation should be needed, but the file makeall.csh can be used to recompile the software if necessary.

The Icy/CSPICE tar package will include the following directories

data	Sample data files used with the "cookbook" routines
doc	Documentation, in both plain-text and HTML formats
exe	Executable utility files
lib	Object libraries and idl.dlm
<pre>src/icycook</pre>	Sample IDL "cookbook" procedures

plus other directories not discussed here. It's instructive to look over the IDL "cookbook" procedures as an introduction to using Icy/CSPICE.

SPICE consists of two parts: the library itself, and a collection of stand-alone utilities in the exe directory. The latter are mainly used for managing the contents of kernel files. For example, the toxfer program converts binary kernel files into an ASCII transfer format, while tobin does the reverse.

Figure 1: Synopsis of the contents of the sample STEREO ephemeris file.

In order to use the Icy/CSPICE library, the SPICE DLM needs to be registered. The simplest way to do this is to use the DLM_REGISTER procedure, e.g.

```
IDL> dlm_register, '/usr/local/icy/lib/idl.dlm'
```

In IDL, all SPICE routines start with the prefice "CSPICE_", for example CSPICE_SPKEZR, CSPICE_SUBPT, etc.

SPICE toolkit version N0058 introduced parameterized dynamic frames, which are used to implement some of the heliospheric coordinate system used by STEREO. Users with earlier versions of the SPICE toolkit should upgrade to N0058 or later.

3 SPICE kernels

The term "kernels" is used to denote data files used by the SPICE system. At the time of writing, the sample STEREO orbit kernel is stereo_20060211a_nominal.xsp with a creation date of February 6, 2004. This is an example of what is called an SPK, or Spacecraft and Planet Kernel. The file extension .xsp means that it's in a platform-independent ASCII transfer format. This needs to be converted into a binary file before it can be used. The tobin utility in the exe directory will convert the ASCII file into a binary file with the extension .bsp. These binary files are platform-specific, but modern versions of SPICE can read binary files from other platforms, with the exception of OpenVMS.

The brief program can be used to get a brief summary of the contents of a binary SPK file. The results for the sample STEREO ephemeris file are shown in Figure 1. This tells us that the file contains ephemerides for two bodies, named "STEREO AHEAD" and "STEREO BEHIND", with ID numbers -234 and -235 respectively. (Spacecraft ID numbers are always negative in SPICE, while planets and other solar system bodies have positive ID numbers.) Also, we have the date and time ranges that the file is valid for. (SPICE uses ephemeris time (ET) which has a small offset from TAI or UTC—this will be discussed in more detail later.) The spacit utility can be used to display more detailed information about the file.

Although the file stereo_20060211a_nominal.bsp contains all the ephemeris data needed for the two STEREO spacecraft, it cannot be used by itself; it needs to be put into its context within the solar system. Specifically, the positions of the two STEREO are defined relative to Earth early in the mission, and relative to the Sun later in the mission, while SPICE internally needs to relate all of these to the solar system barycenter. Thus, a solar system ephemeris needs to be loaded. There are several generic solar system kernels available, but the correct one to use is de405.bsp, because that was the kernel that was used to generate the STEREO ephemeris.

Another data file required by SPICE is a Leapseconds Kernel (LSK), to support various time conversions. The most recent leapseconds file is naif0007.tls.

A Planetary Constants Kernel (PCK) allows coordinates to be expressed in longitude and latitude. For example, pck00007.tpc is a fairly modern and comprehensive PCK for the solar system as a whole. Using this file allows one to extract either geographic or heliographic (i.e. Carrington) coordinates.

All of the generic kernels discussed above can be downloaded from the NAIF website. Also, the STEREO Science Center will organize and make available all SPICE kernels needed to process the orbit and attitude data, including the STEREO-specific and relevant NAIF generic kernels.

4 Basic software calls

4.1 Loading SPICE kernels

SPICE kernels are loaded with calls to CSPICE_FURNSH, one for each kernel file. For example, to load the STEREO ephemeris, together with the generic planet ephemeris and leapseconds, one would type

```
cspice_furnsh, 'naif0007.tls'
cspice_furnsh, 'de405.bsp'
cspice_furnsh, 'stereo_20060211a_nominal.bsp'
```

Assuming that each of these files are in the current directory. Similarly, calls to CSPICE_UNLOAD can be used to unload kernels once they are no longer needed.

4.2 Time

Internally, SPICE uses Ephemeris Time (ET), also known as Barycentric Dynamical Time (TDB). This time varies from TAI time by 32.184 seconds plus a small annual variation on the order of a few milliseconds. In SPICE, ET is expressed as the number of dynamical seconds since 1-Jan-2000 12:00 ET.

The routine CSPICE_STR2ET converts a wide variety of time string formats into double precision ET values. A simple way to convert time values into ET format is to first use the SolarSoft routine ANYTIM2UTC with the /CCSDS keyword to convert date/time values into ISO 8601 format, which SPICE is able to ingest, e.g.

```
utc = anytim2utc(date_time, /ccsds)
cspice_str2et, utc, et
```

Table 1: NAIF ID names and numbers for various solar system objects, and the two STEREO spacecraft. Character string names are case-insensitive.

MERCURY BARYCENTER		SATURN BARYCENTER	6	MERCURY	199
VENUS BARYCENTER	2	URANUS BARYCENTER	7	VENUS	299
EARTH BARYCENTER	3	NEPTUNE BARYCENTER	8	MOON	301
MARS BARYCENTER		PLUTO BARYCENTER	9	EARTH	399
JUPITER BARYCENTER	5	SUN	10	MARS	499
STEREO AHEAD -234		SOLAR SYSTEM BARYCENTER (or SSB)			0
STEREO BEHIND -235		EARTH MOON BARYCENTER (or EMB)			3

(The CCSDS keyword refers to the Consultative Committee for Space Data Systems, which endorsed a specific subset of the wider ISO-8601 standard.)

Similarly, the routine CSPICE_ET2UTC can be used to convert from ephemeris time into a UTC character string. For example, the call

```
cspice_et2utc, et, 'ISOC', 3, utcstr
```

would return a string such as "2000-01-01T12:00:00.000", which is easily ingested into SolarSoft. The "ISOC" tells the software to use ISO 8601 calendar format, and the "3" tells the software to return seconds to three significant figures past the decimal point.

The above SPICE calls allow the input times to be either scalars or one-dimensional arrays.

4.3 Spacecraft and planet identifiers

Spacecraft and planets are identified within SPICE in one of two ways: either by a character string name, or by an ID number. Table 1 shows the ID numbers for various solar system objects, and the two STEREO spacecraft. By convention, spacecraft IDs are always negative.

Internally, SPICE uses ID numbers to track spacecraft and solar system objects, and lower level routines take the ID numbers as input parameters. Higher level routines, however, use character string names, such as "Earth" or "Sun". A few high level routines can take either the name or the ID number, provided that the ID number is formatted as a character string.

4.4 Reference frames

SPICE incorporates two basic kinds of reference frames: inertial and body-fixed frames. Examples of inertial frames include:

J2000: Celestial reference frame, with Z aligned with Earth's rotational axis, and X aligned with the direction of the vernal equinox, at the dynamical epoch of J2000.

ECLIPJ2000: Ecliptic coordinates based upon the J2000 frame.

among others. Note that these inertial frames are based on the Earth's rotational axis and ecliptic for a specific date, and do not take precession into account. This is discussed in more detail in Appendix A.

Body-fixed frames include geographic and Carrington heliographic, as well as spacecraft and instrument-based frames. The reference frames for geographic and Carrington heliographic coordinates are IAU_EARTH and IAU_SUN respectively.

Versions N0058 and later of the SPICE toolkit also allows one to define additional non-inertial reference frames based on certain kinds of parameters. This facility is used to realize certain standard heliospheric coordinate systems, discussed below.

The routine CSPICE_SPKEZR provides a simple interface to extract coordinate frame data of objects. The calling sequence is

```
CSPICE_SPKEZR, TARGET, ET, REF, ABCORR, OBS, STATE, LTIME
```

where the input parameters are

TARGET: Target body name.

ET: Ephemeris time(s) of the observation.

REF: Reference frame, e.g. 'ECLIPJ2000' or 'IAU_SUN'.

ABCORR: Aberration correction. Possible values include:

'NONE' No aberration correction

'LT' Correct for one-way light travel time

'LT+S' Correct for one-way light travel time, and stellar aberration

One can also use 'XLT' and 'XLT+S' in the transmission case, where ET then refers to the time of transmission, and ET+LTIME is the observation time.

OBS: The name of the observing body; i.e. the origin of the coordinate system.

and the output parameters are

STATE: The six-element state vector of the target. The first three parameters are the position in kilometers, and the last three are the velocity in km/sec. (Note that velocities expressed in rotating coordinate frames have different magnitudes than velocities expressed in inertial coordinate frames.)

LTIME: The one-way light travel time from the target to the observer.

The target and observer parameters can be either their character string names, or their ID numbers formatted as a character string. For example, the call

```
cspice_spkezr, '-234', et, 'J2000', 'LT+S', 'Earth', state, ltime
```

returns the state vector of the STEREO Ahead spacecraft as seen from Earth in the J2000 celestial coordinate system. The complete process from loading to unloading the kernels is presented in Figure 2.

```
pro stereo_a_demo, date, state, ltime
;
cspice_furnsh, 'naif0007.tls'
cspice_furnsh, 'de405.bsp'
cspice_furnsh, 'pck00007.tpc'
cspice_furnsh, 'stereo_20060211a_nominal.bsp'
;
utc = anytim2utc(date,/ccsds)
cspice_str2et, utc, et
;
cspice_spkezr, 'STEREO AHEAD', et, 'IAU_SUN', 'None', 'Sun', state, ltime
;
cspice_unload, 'stereo_20060211a_nominal.bsp'
cspice_unload, 'pck00007.tpc'
cspice_unload, 'de405.bsp'
cspice_unload, 'naif0007.tls'
;
end
```

Figure 2: Sample IDL program for deriving the position and velocity of the STEREO A spacecraft in Carrington heliographic coordinates at a given date.

5 Coordinate systems

It is possible to use the SPICE software to derive coordinates in a number of standard coordinate systems, described below. Two axes are chosen to define each coordinate system, and the missing axis satisfies the right-hand rule. Table 2 gives the values for the REF and OBS parameters in the CSPICE_SPKEZR call.

Some coordinate systems require that an appropriate parameterized dynamic frame be loaded. These dynamic frames are defined in the files heliospheric.tf, and in stereo_rtn.tf.

- **GEI:** Geocentric Equatorial Inertial. The X axis points toward the first point of Aries (i.e. the vernal equinox), and the Z axis is aligned with the geographic north pole. When expressed in terms of longitude and latitude, this is the well known celestial coordinate system of right ascension and declination. This is realized with the J2000 ecliptic.
- **GEO:** Geographic. X is the intersection of the Greenwich meridian and the geographic equator, and Z is the geographic North Pole. This is usually expressed as geographic longitude and latitude. (However, see Appendix B for a discussion of the distinction between planetographic and planetocentric coordinates.)
- **GSE:** Geocentric Solar Ecliptic. X is the Earth-Sun line, and Z is aligned with the north pole for the ecliptic of date.
- **HCI:** Heliocentric Inertial. Z is the solar north rotational axis, and X is the solar ascending node on the J2000 ecliptic.

Table 2: REF and OBS values to be used in CSPICE_SPKEZR to realize various standard coordinate systems. Coordinate systems marked with the "J2000" subscript are referenced to the ecliptic plane at the J2000 epoch; those marked with the "D" subscript are referenced to the ecliptic of date. An asterisk (*) means that a dynamic frame needs to be loaded.

System			REF	OBS
GEI	J2000		J2000	Earth
GEO			IAU_EARTH	Earth
GSE	D	*	GSE	Earth
HCI	J2000	*	HCI	Sun
$_{ m HAE}$	J2000		ECLIPJ2000	Sun
$_{ m HAE}$	D	*	ECLIPDATE	Sun
HEE	D	*	$_{ m HEE}$	Sun
HEEQ		*	HEEQ	Sun
Carrington			IAU_SUN	Sun
STEREO-A RTN		*	STAHGRTN	Sun (or STEREO Ahead)
STEREO-B RTN		*	STBHGRTN	Sun (or STEREO Behind)

HAE: Heliocentric Aries Ecliptic. X is in the direction of the first point of Aries, and Z is the ecliptic North Pole. This can be realized with either the J2000 ecliptic, or the ecliptic of date.

HEE: Heliocentric Earth Ecliptic. X is the Sun-Earth line, and Z is the north pole for the ecliptic of date.

HEEQ: Heliocentric Earth Equatorial. Z is the solar rotation axis, and X is in the plane containing the Z axis and Earth, at the intersection of the solar central meridian, and the heliographic equator. When converted to longitude and latitude, this is known as Stonyhurst heliographic coordinates.

Carrington heliographic. Z is the solar rotational axis. The X axis is at the intersection of the Carrington prime meridian and the heliographic equator. The coordinate system rotates with a sidereal period of 25.38 days. This coordinate system is usually expressed in terms of longitude and latitude.

HGRTN/RTN: Radial-Tangential-Normal. X axis points from Sun center to the spacecraft, and the Y axis is the cross product of the solar rotational axis and X, and lies in the solar equatorial plane (towards the West limb). For the STEREO Ahead spacecraft, this is realized through the dynamic coordinate frame STAHGRTN, while for STEREO Behind it is realized through STBHGRTN. When the Sun is used as the origin, the designation is HGRTN—with the spacecraft as origin, it's simply RTN. RTN coordinates converted to longitude and latitude are also known as helioprojective-cartesian (see Thompson (2005)).

5.1 Longitude and latitude

Any of the above coordinate systems can be converted into longitude and latitude, and vice-versa, by using the CSPICE calls

```
cspice_reclat, vector, radius, lon, lat
cspice_latrec, radius, lon, lat, vector
```

Where VECTOR is a three-element array, e.g. either the position or velocity portion of the STATE vector. These are equivalent to the IDL calls

respectively. An advantage of using the SPICE calls is that the parameters can be vectorized: VECTOR can be a $3 \times N$ array, and RADIUS, LON, and LAT can each be arrays of N elements.

SPICE, like IDL, expresses angles in radians. Therefore, a conversion is needed if degrees or arc seconds are desired. The SPICE calls CSPICE_DPR() and CSPICE_RPD() return the double-precision values $180/\pi$ and $\pi/180$ respectively. For example, to convert the output of CSPICE_RECLAT from radians to degrees, one could use the commands

```
lon = lon * cspice_dpr()
lat = lat * cspice_dpr()
```

6 Coordinate conversions

Because all coordinate systems are implemented as frames, either internal or dynamic, the process to convert from one coordinate system to another is the same for all. There are two basic routines used for converting between coordinate frames, CSPICE_SXFORM and CSPICE_PXFORM. The only difference between them is that CSPICE_SXFORM works on 6-element state vectors containing both position and velocity, which CSPICE_PXFORM works only on 3-element position vectors. The parameters are:

FROM: Name of the reference frame one is converting from.

TO: Name of the reference frame one is converting to.

ET: Ephemeris time(s) of the observation.

XFORM: Output transformation matrix. If ET is an array, then this will be either a $6 \times 6 \times N$ array, or a $3 \times 3 \times N$ array, depending on whether one is calling CSPICE_SXFORM or CSPICE_PXFORM.

For the FROM and TO parameters, use the REF field from Table 2 for the coordinate systems in question.

The following example demonstrates the conversion from GEI to HEEQ coordinates, which involves both rotation of the coordinate axes, and a shift of the shift of the origin. Note that both internal and dynamic reference frames are involved.

```
cspice_spkezr, 'STEREO AHEAD', et, 'J2000', 'None', 'Earth', gei, ltime
cspice_spkezr, 'Sun', et, 'J2000', 'None', 'Earth', gei_sun, ltime_earth
cspice_sxform, 'J2000', 'HEEQ', et, xform
heeq = transpose(xform) # (gei - gei_sun)
```

References

- Fränz, M. and Harper, D. (2002). Heliospheric coordinate systems. *Planetary and Space Science*, **50**, 217–239.
- Hapgood, M. A. (1992). Space physics coordinate transformation: A user guide. *Planetary and Space Science*, **40**, 711–717.
- Russell, C. T. (1971). Geophysical coordinate transformations. Cosmic Electrodynamics, 2, 184–196.
- Thompson, W. T. (2005). Coordinate systems for solar image data. *Astron. Astrophys.* in prep., http://orpheus.nascom.nasa.gov/~thompson/papers/coordinates.ps.

Appendices

A Precession

Internally, SPICE works within the J2000 inertial reference frame. However, Earth's rotational axis slowly precesses around the pole of the ecliptic at a rate of about 50" per year, with a period of about 26,000 years. Also, the plane of the ecliptic itself precesses at a rate of about 47" per century. There may be situations where these effects need to be taken into account.

Fränz and Harper (2002), discuss the effects of precession on the standard heliospheric coordinate systems, and distinguish between three different versions of the Geocentric Equatorial Inertial system:

 GEI_{J2000} Earth mean equator at J2000.

GEI_D Earth mean equator of date, i.e. corrected for precession.

GEI_T Earth true equator of date, i.e. corrected for precession and nutation.

SPICE returns GEI_{J2000} . On the other hand, geographic coordinates (GEO) are defined relative to the true equator of date, which is how they are implemented in SPICE.

Similarly, Franz and Harper distinguish between two versions of Heliocentric Aries Ecliptic:

HAE_{J2000} Earth mean ecliptic at J2000. HAE_D Earth mean ecliptic of date.

Again, SPICE natively returns HAE_{J2000} . However, the other ecliptic systems, GSE and HEE, are explicitly defined by Fränz and Harper in terms of the mean ecliptic of date, while SPICE wants to work within the J2000 reference frame. The use of parameterized dynamic frames, introduced in SPICE toolkit version N0058, gets around this problem. Table 2 lists which coordinate systems are based on the J2000 ecliptic, and which use the ecliptic of date.

Precession is not a concern for HEEQ, RTN, or Carrington heliographic coordinates.

B Planetographic versus planetocentric coordinates

In cartography, a distinction is made between planetocentric and planetographic coordinates. Planetocentric latitude is the angle between the equatorial plane and a line drawn from the center of mass to the reference point. Planetographic latitude, however, is defined in terms of a line which is normal to a spheroidal reference surface for the body in question. For spherical bodies, such as the Sun, there is no distinction between these two kinds of latitudes, but there is a difference for an oblate spheroid such as Earth. For the STEREO spacecraft, the distinction between geocentric and geographic latitude varies between about $\pm 0^{\circ}.15$. The coordinates returned by the CSPICE_SPKEZR call in Section 5 are in the planetocentric system.

There's also a distinction between planetocentric and planetographic longitude when the body's rotation is retrograde, but this does not apply to either Earth or the Sun.

Fränz and Harper (2002) explicitly define the GEO coordinate system as a planetographic system. However, Russell (1971) and Hapgood (1992) describe the GEO system in terms of X, Y, Z coordinates, where the distinction between planetocentric and planetographic does not apply. One should therefore be careful when comparing geographic longitudes and latitudes to X, Y, Z systems.

Although SPICE returns planetocentric coordinates, the routines CSPICE_BODVAR and CSPICE_RECGEO can be used to derive planetographic longitude and latitude (in degrees), as well as the altitude above the surface, e.g.

```
cspice_spkezr, 'STEREO Ahead', et, 'IAU_EARTH', abcorr, 'Earth', state, ltime
cspice_bodvar, 399, 'RADII', radii
flat = (radii[0]-radii[2]) / radii[0]
cspice_recgeo, state[0:2], radii[0], flat, lon, lat, alt
lon = lon * cspice_dpr()
lat = lat * cspice_dpr()
```

Note that CSPICE_BODVAR requires the body ID number, and that 'RADII' must be uppercase.

The inverse routine is CSPICE_GEOREC, e.g.

```
lon_r = lon * cspice_rpd()
lat_r = lat * cspice_rpd()
cspice_georec, lon_r, lat_r, alt, radii[0], flat, vector
```

where LON_R and LAT_R are the longitude and latitude converted back into radians, and VECTOR is the output position vector.

The parameters can be vectorized, with VECTOR being a $3 \times N$ array, and LON, LAT, and ALT each being arrays of N elements.

C Using CSPICE_SUBPT

An alternative way to derive heliographic coordinates is to use CSPICE_SUBPT, which returns the sub-observer point on a spheroidal body. The advantage of using it over CSPICE_SPKEZR is that CSPICE_SUBPT treats the STEREO spacecraft as an observer rather than as a target, which influences how aberration corrections are handled. The calling sequence is

```
CSPICE_SUBPT, METHOD, TARGET, ET, ABCORR, OBS, SPOINT, ALT
```

where the input parameters are

METHOD: Calculation method. Possible values are:

'NEAR POINT' Nearest point on surface to observer 'INTERCEPT' Intercept point along line from target center

For a spherical object like the Sun, the two methods return the same result. The 'INTERCEPT' method is less computationally intensive.

TARGET: Target body name, e.g. 'Sun'.

ET: Ephemeris time of the observation (scalar).

ABCORR: Aberration correction. (See Section 4.4.)

OBS: The name of the observing body, e.g. 'STEREO AHEAD'.

and the output parameters are

SPOINT: Planetocentric coordinates of the sub-observer point on the target surface.

ALT: The altitude, calculated as the distance from the sub-observer point to the observer.

The following example shows the derivation of Carrington heliographic longitude and latitude (in degrees) using CSPICE_SUBPT

```
cspice_subpt, 'Intercept', 'Sun', et, 'lt+s', 'STEREO Ahead', spoint, alt
cspice_reclat, spoint, radius, lon, lat
lon = lon * cspice_dpr()
lat = lat * cspice_dpr()
```

The radius returned by CSPICE_RECLAT will be the solar radius of 696,000 km, and the sum radius+alt will be the heliocentric distance. Without aberration correction, the results will be identical with those from CSPICE_SPKEZR.